



SAS で Forest Plot

舟尾 暢男



使用するデータ:*sashelp.iris*


SepalLength	SepalWidth	PetalLength	PetalWidth	Species
5.1	3.5	1.4	0.2	Setosa
4.9	3.0	1.4	0.2	Setosa
4.7	3.2	1.3	0.2	Setosa
4.6	3.1	1.5	0.2	Setosa
5.0	3.6	1.4	0.2	Setosa
5.4	3.9	1.7	0.4	Setosa
4.6	3.4	1.4	0.3	Setosa
...

- フィッシャーが判別分析法を紹介するために利用したアヤメの品種分類（変数 Species : Setosa、Versicolor、Virginica）に関するデータ
⇒ 以下の 4 変数を説明変数としてアヤメの種類を判別しようとした

- SepalLength : アヤメのがくの長さ
- SepalWidth : アヤメのがくの幅
- PetalLength : アヤメの花弁の長さ
- PetalWidth : アヤメの花弁の幅



Graphic by (c)Tomo.Yun (<http://www.yunphoto.net>)



MYDATA


```

ods listing close ;
proc means data=sashelp.iris; by Species; var SepalLength; output out=TMP1 mean=M P1=L P99=U ;
proc means data=sashelp.iris; by Species; var SepalWidth ; output out=TMP2 mean=M P1=L P99=U ;
proc means data=sashelp.iris; by Species; var PetalLength; output out=TMP3 mean=M P1=L P99=U ;
proc means data=sashelp.iris; by Species; var PetalWidth ; output out=TMP4 mean=M P1=L P99=U ;
run ; ods listing ;

data MYDATA(drop=_TYPE_ Species rename=(_FREQ_=N)) ;
  set TMP1 TMP2 TMP3 TMP4 TMP1 TMP2 TMP3 TMP4 ;
  ID=_N_ ; Group="Group "||strip(_N_) ; M=round(M,0.1) ;
  if (Species="Setosa") then do ; R=Group ; W=0 ; end ;
  else W=1 ;
  label M="Mean" L="Lower" U="Upper" ;
run ;

```

ID	Group	N	M	L	U	R	W
1	Group 1	50	50.1	43	58	Group 1	0
2	Group 2	50	59.4	49	70		1
3	Group 3	50	65.9	49	79		1
4	Group 4	50	34.3	23	44	Group 4	0
5	Group 5	50	27.7	20	34		1
6	Group 6	50	29.7	22	38		1
7	Group 7	50	14.6	10	19	Group 7	0
8	Group 8	50	42.6	30	51		1
9	Group 9	50	55.5	45	69		1
10	Group 10	50	2.5	1	6	Group 10	0
11	Group 11	50	13.3	10	18		1
12	Group 12	50	20.3	14	25		1




例1: Forest Plot

```
ods listing gpath="c:/temp" image_dpi=300 ;
ods graphics / reset height=450px width=600px ;



proc sgplot data=MYDATA nowall noborder noautolegend ;
    highlow y=GROUP low=L high=U
        / lineattrs=(pattern=1 color=blue thickness=2) ;
    scatter y=Group x=M
        / markerattrs=(symbol=circlefilled size=10 color=red) ;

    yaxis display=(noticks nolabel)
        reverse offsetmin=0.1 offsetmax=0.1 ;
    xaxis display=(noticks nolabel)
        values=(0 to 80 by 20) offsetmin=0.1 ;
run ;
```

- 線種:highlow ステートメントの lineattrs にて
- 点種:scatter ステートメントの markerattrs にて
- scatter ステートメントの xerrorupper, xerrorlower, errorbarattrs
はエラーバーの属性制御が困難な場合があるので使用せず




例1: Forest Plot



線の指定

番号	キーワード	線
1	Solid	—————
2	ShortDash	- - - - -
4	MediumDash	— — — — —
5	LongDash	—— — — — —
8	MediumDashShortDash	— - — - — - — -
14	DashDashDot	- - - - - - - -
15	DashDotDot	- . - - - - - - -
20	Dash	- - - - - - - -
26	LongDashShortDash	— — - - — - — -
34	Dot
35	ThinDot	- - - - - - - -
41	ShortDashDot	- - - - - - - -
42	MediumDashDotDot	— - — - — - — -



点の指定


点	ラベル	点	ラベル	点	ラベル	点	ラベル
↓	ArrowDown	I	Ibeam	<	TriangleLeft	▼	HomeDownFilled
*	Asterisk	+	Plus	>	TriangleRight	■	SquareFilled
○	Circle	□	Square	U	Union	★	StarFilled
◇	Diamond	☆	Star	X	X	▲	TriangleFilled
>	GreaterThan	T	Tack	Y	Y	▼	TriangleDownFilled
<	LessThan	~	Tilde	Z	Z	◀	TriangleLeftFilled
#	Hash	△	Triangle	●	CircleFilled	▶	TriangleRightFilled
▼	HomeDown	▽	TriangleDown	◆	DiamondFilled		

例2: Forest Plot の点を装飾



```
proc sgplot data=MYDATA nowall noborder noautolegend ;
    highlow y=GROUP low=L high=U
        / lineattrs=(pattern=1 color=blue thickness=2) ;
    scatter y=Group x=M
        / filledoutlinedmarkers
            markeroutlineattrs=(color=red thickness=1)
            markerfillattrs=(color=yellow)
            markerattrs=(symbol=circlefilled size=10) ;

    yaxis display=(noticks nolabel)
        reverse offsetmin=0.1 offsetmax=0.1 ;
    xaxis display=(noticks nolabel)
        values=(0 to 80 by 20) offsetmin=0.1 ;
run ;
```

- プロット点について「線」と「塗りつぶし」の色等を別々に指定したい場合は、filledoutlinedmarkers 付きで属性制御
- 何気に yaxis ステートメントの reverse が便利



例2: Forest Plot の点を装飾




例3: Forest Plot に各種情報をテキスト表示

```
proc sgplot data=MYDATA nowall noborder nocycleattrs noautolegend ;
    highlow    y=GROUP low=L high=U / lineattrs=(pattern=1 color=blue thickness=2) ;
    scatter     y=Group x=M / markerattrs=(symbol=circlefilled size=10 color=red) ;
    yaxistable Group N M L U / y=Group location=inside position=left ;



    yaxis display=none reverse offsetmin=0.1 offsetmax=0.1 ;
    xaxis display=(noticks nolabel) values=(0 to 80 by 20) offsetmin=0.1 ;
run ;
```

- 例 1 を改変
- yaxistable で各種情報をテキストで出力
 - 群情報も yaxistable で出力
 - よって yaxis にて display=none を設定

※ 巷では scatter ステートメントでハザード比等の情報を出力している資料もあるが、時代遅れ



例3: Forest Plot に各種情報をテキスト表示




例4: Forest Plot に参照線や注釈を追記

```
proc sgplot data=MYDATA nowall noborder nocycleattrs noautolegend ;
    highlow    y=GROUP low=L high=U / lineattrs=(pattern=1 color=blue thickness=2) ;
    scatter     y=Group x=M / markerattrs=(symbol=circlefilled size=10 color=red) ;
    yaxistable Group N / y=Group location=inside position=left ;
    yaxistable M L U   / y=Group location=inside position=right ;



    refline 20 / axis=x lineattrs=(pattern=dash color=green) transparency=0.5 ;
    inset '          <20mm' / position=bottomleft ;
    inset '>20mm'           / position=bottom ;

    yaxis display=none reverse offsetmin=0.1 offsetmax=0.1 ;
    xaxis display=(noticks nolabel) values=(0 to 80 by 20) offsetmin=0.1 ;
run ;
```

- 例 3 を改変
- yaxistable を分割
- 参照線 x=20 を追記
- inset ステートメントで注釈を追記




例4: Forest Plot に参照線や注釈を追記




例5: Forest Plot をとりあえず完成させる

```
proc sgplot data=MYDATA nowall noborder nocycleattrs noautolegend ;  
    highlow    y=GROUP low=L high=U / lineattrs=(pattern=1 color=blue thickness=2) ;  
    scatter     y=Group x=M / markerattrs=(symbol=circlefilled size=10 color=red) ;  
    yaxistable Group / y=Group location=inside position=left labelattrs=(size=8)  
                        valueattrs=(size=6) indentweight=W ;  
    yaxistable M L U / y=Group location=inside position=right ;  
  
    refline R / axis=y lineattrs=(thickness=20 color=yellow transparency=0.5) ;  
    refline 20 / axis=x lineattrs=(pattern=dash color=green transparency=0.5) ;  
    inset '          <20mm' / position=bottomleft ;  
    inset '>20mm'           / position=bottom ;  
  
    yaxis display=none reverse offsetmin=0.1 offsetmax=0.1 ;  
    xaxis display=(noticks nolabel) values=(0 to 80 by 20) offsetmin=0.1 ;  
run ;
```

- Group の種類によってインデントを追加(変数 W)
- Group の種類によってグラフに色を塗る(変数 R)




例5: Forest Plot をとりあえず完成させる



sgplot プロジェクト vs GTL

- sgplot プロジェクトだけではカスタマイズが困難な場合が…
 - テキスト部分とプロット部分の幅の割合
 - マニアックな指定
- そこで Graph Template Language (GTL) を使用




例6: GTL で Forest Plot を作成

```
proc template;
define statgraph mytemplate ;
begingraph ;      * 列の数と各列の幅を指定 ;
layout lattice / columns=6 columnweights=(0.15 0.1 0.45 0.1 0.1 0.1) ;
*** 1列目：カテゴリ ;
layout overlay / walldisplay=none yaxisopts=(reverse=true display=none) ;
axistable y=Group value=Group ;
endlayout ;

*** 2列目：例数 ;
layout overlay / walldisplay=none yaxisopts=(reverse=true display=none) ;
axistable y=Group value=N ;
endlayout ;

*** 3列目：プロット ;
layout overlay / walldisplay=none yaxisopts=(reverse=true display=none)
xaxisopts=(label='<20mm                                >20mm'
linearopts=(tickvaluepriority=true tickvaluelist=(0 20 40 60 80))) ;
scatterplot    y=Group x=M / markerattrs=(symbol=circlefilled) ;
highlowplot   y=Group low=L high=U ;
referenceline x=20 / lineattrs=(pattern=dash) ;
endlayout;
```



例6: GTL で Forest Plot を作成


```
*** 4列目：平均値 ;
layout overlay / walldisplay=none yaxisopts=(reverse=true display=none) ;
axistable y=Group value=M ;
endlayout;

*** 5列目：エラーバー（下側） ;
layout overlay / walldisplay=none yaxisopts=(reverse=true display=none) ;
axistable y=Group value=L ;
endlayout;



*** 6列目：エラーバー（上側） ;
layout overlay / walldisplay=none yaxisopts=(reverse=true display=none) ;
axistable y=Group value=U ;
endlayout;
endlayout ;
endgraph ;
end ;

run ;

proc sgrender data=MYDATA template=mytemplate ;
run ;
```



例6: GTL で Forest Plot を作成




例7: GTL で Forest Plot を作成(もう少し加工)

```
proc template;
define statgraph mytemplate ;
dynamic _BW _COL _DT ;
begingraph ;
layout lattice / columns=6 columnweights=(0.15 0.1 0.45 0.1 0.1 0.1) ;

*** ヘッダ ;
sidebar / align=top ;
layout lattice / rows=1 columns=5 columnweights=(0.15 0.1 0.43 0.1 0.22) ;
entry textattrs=(size=8) halign=left "Group" ;
entry textattrs=(size=8) halign=left "# of Sub." ;
entry textattrs=(size=8) halign=center "Mean Plot" ;
entry textattrs=(size=8) halign=left "Mean" ;
entry textattrs=(size=8) halign=center "Percentile " ;
endlayout ;
endsidebar ;

*** 1列目：カテゴリ ;
layout overlay / walldisplay=none yaxisopts=(reverse=true display=none) ;
referenceline y=R / lineattrs=(thickness=_BW color=_COL)
datatransparency=_DT ;
axistable      y=Group value=Group / valueattrs=(color=blue size=10
family="Arial Unicode MS" weight=bold) indentweight=W label=" " ;
endlayout ;
```




例7: GTL で Forest Plot を作成(もう少し加工)

```
*** 2列目: 例数 ;
layout overlay / walldisplay=none yaxisopts=(reverse=true display=none) ;
referenceline y=R / lineattrs=(thickness=_BW color=_COL)
datatransparency=_DT ;
axistable      y=Group value=N / valueattrs=(color=blue size=8) label=" " ;
endlayout;

*** 3列目: プロット ;
layout overlay / walldisplay=none yaxisopts=(reverse=true display=none)
xaxisopts=(label=" " linearopts=(tickvaluepriority=true
tickvaluelist=(0 20 40 60 80))) ;
referenceline y=R / lineattrs=(thickness=_BW color=_COL)
datatransparency=_DT ;
scatterplot    y=Group x=M / markerattrs=(symbol=circlefilled) ;
highlowplot   y=Group low=L high=U ;
referenceline x=20 ;

drawarrow x1=7 y1=2.5 x2=22 y2=2.5 / drawspace=layoutpercent
arrowheaddirection=in ;
drawarrow x1=77 y1=2.5 x2=92 y2=2.5 / drawspace=layoutpercent
arrowheaddirection=out ;
drawtext "Small" / x=23 y=2.5 width=20 drawspace=layoutpercent
anchor=left border=false ;
drawtext "Large" / x=76 y=2.5 width=20 drawspace=layoutpercent
anchor=right border=false ;
endlayout;
```




例7: GTL で Forest Plot を作成(もう少し加工)


```
*** 4列目: 平均値 ;
layout overlay / walldisplay=none yaxisopts=(reverse=true display=none) ;
referenceline y=R / lineattrs=(thickness=_BW color=_COL)
datatransparency=_DT ;
axistable      y=Group value=M / display=(label) label=" ";
endlayout;

*** 5列目: エラーバー (下側) ;
layout overlay / walldisplay=none yaxisopts=(reverse=true display=none) ;
referenceline y=R / lineattrs=(thickness=_BW color=_COL)
datatransparency=_DT ;
axistable      y=Group value=L / display=(label) label="1%"
labelattrs=(color=green size=10 weight=bold) ;
endlayout;


*** 6列目: エラーバー (上側) ;
layout overlay / walldisplay=none yaxisopts=(reverse=true display=none) ;
referenceline y=R / lineattrs=(thickness=_BW color=_COL)
datatransparency=_DT ;
axistable      y=Group value=U / display=(label) label="99%"
labelattrs=(color=green size=10 weight=bold) ;
endlayout;
endlayout ;
endgraph ;
end ;
run ;
proc sgrender data=MYDATA template=mytemplate ;
dynamic _BW=20 _COL="cyan" _DT=0.8 ; run ;
```



例7: GTL で Forest Plot を作成(もう少し加工)



※ 文字のフォントについて : [Using Fonts with Universal Printers and SAS/GRAPH Devices](#)



参考文献

- SAS "Graph Template Language Reference", Fifth Edition.
<http://support.sas.com/documentation/cdl/en/grstatgraph/69718/HTML/default/viewer.htm>
- SAS "Graph Template Language Tip Sheet".
http://support.sas.com/rnd/app/ODSGraphics/TipSheet_GTL.pdf
- SAS/STAT User's Guide "Customizing the Kaplan–Meier Survival Plot".
<https://support.sas.com/documentation/onlinedoc/stat/141/kaplan.pdf>
<https://support.sas.com/documentation/onlinedoc/stat/151/kaplan.pdf>
- Sanjay Matange (2013) "Getting Start with the Graph Template Language in SAS", SAS Institute.
- SAS Graph Template Language 超入門
http://nfunao.web.fc2.com/files/Introductio_n_to_SAS_GTL.pdf

